
Working Smarter

Three COBOL Tips for
Day-to-Day Programming

This month, I will present three COBOL
tips that can be used in day-to-day

commercial programming. The first two
require at least a level of COBOL II V1R3,
whereas the third requires COBOL for
MVS (Language Environment COBOL).

TIP #1: DYNAMICALLY
DETERMINING THE NUMBER
OF ENTRIES IN A TABLE 

In COBOL, when an array (table) is
declared with an OCCURS clause, the array
must always be assigned a maximum number
of entries. This is coded as either:

1. OCCURS 10 TIMES

or 

2. OCCURS 0 TO 100 TIMES DEPENDING ON
identifier-1

In both cases, the compiler reserves the
required amount of memory to accommo-
date a full-size table.

The Hardcode Problem
How does a programmer communicate to

the Procedure Division the maximum number
of entries in a given OCCURS structure? I
refer to this as the "hardcode problem."
Figure 1 shows a sample bit of code that
demonstrates a technique that has been
used for years as a way to circumvent the
problem. That is, an elementary item is
associated with an OCCURS. It is assigned
a VALUE clause number that matches the
OCCURS number.

What happens all too frequently is that
a maintenance programmer inadvertently
forgets to change the VALUE clause
number to match the OCCURS number
when the table increases (or decreases) in
number of entries.

Make it Dynamic
By using the LENGTH OF special register,

all top-of-table variables can be dynamically
computed. Maintenance, in the form of
OCCURS number changes, will be auto-
matically accounted for when this technique
is used. See Figure 1 and note the following:

● The COMPUTE statement that
calculates MAX-JCL-ENTRIES
would be best placed in the 
initialization ("housekeeping")
routine of the program.

● All top-of-table variables should be
fullword aligned (01 levels) with
Packed-Decimal usage.

● A unique variable should be supplied
for every OCCURS where top-of-table
is an issue.

TIP #1 SUMMARY

The technique works because at compile-
time, the LENGTH OF special register is
available for every data item within a
COBOL program, even tables. In Figure 1,
you can visualize what the COMPUTE is
doing by viewing it as:

COMPUTE MAX-JCL-ENTRIES = 240 / 80.

No matter what the OCCURS number is
changed to, by dynamically determining the

number of entries, the COBOL program will
always have an accurate top-of-table value.

TIP #2: HANDLING RUN-TIME
PARAMETERS

The use of run-time parameters is recom-
mended in any language, on any platform.
Using run-time parameters allows a program
to serve multiple purposes. This adds
flexibility and power to an application.
Isn't it easier to change a parameter value
than to write a program?

Parameter Problems in COBOL
When a run-time parameter is passed to a

mainline COBOL program, many times,
the length of the parameter text will vary.
However, just as with the OCCURS clause,
the maximum length must be provided for
within the LINKAGE SECTION. In Figure 2,
the longest length that the parameter value
could have is 12 bytes — PIC X(12) on the
PARM-VALUE field.

In reality, there may be instances where
the value passed is shorter than 12 bytes.
For example, six or three or even a single
byte might be an acceptable parameter
value. In cases such as these, the only
rule is that some parameter value must
always be present (have a length greater
than zero).

01  JCL-Table.
05             Pic  X(80) Value '//jobcard'.
05             Pic  X(80) Value '//STEP1 EXEC PROC=ABC'.
05             Pic  X(80) Value '//EOF*'.

01                  Redefines JCL-Table.
05  JCL-Line Occurs 3 Times,

Indexed by JCL-Index,
Pic  X(80). 

01  Max-JCL-Entries Pic S9(07) Packed-Decimal Value +3.

Procedure Division Code to dynamically calculate number of entries in OCCURS.  

Compute Max-JCL-Entries = 
Length of JCL-Table / Length of JCL-Line (1).

FIGURE 1: WORKING-STORAGE OCCURS AND AN ASSOCIATED "TOP-OF-TABLE" FIELD

In other programs, the requirement might
be that all bytes of the parameter must
always be accounted for on every invocation
of the program. 

In the former situation, where a parameter
value can have any length greater than zero
and less than or equal to the maximum
allowed, a statement such as the following:
"Move Parm-Value To WS-Value"

could move unprintable characters (usually
binary zeros or LOW-VALUES) to the field
named WS-VALUE if the actual parameter
length was shorter than the allotted length
of the LINKAGE SECTION field. These
unwanted characters would appear in bytes
to the right — or trailing.

Use the Provided Length, LENGTH OF
and Reference Modification 

When a parameter is passed via the PARM
keyword in JCL, a halfword length will always
be available just before the actual parameter
text. This is also true when using an ISPF
SELECT with the PGM/PARM keywords as
well as TSO CALL statements. This is a rare
example of actually knowing the length of a
string without having to do anything.

By combining the provided length with
other COBOL features, parameter editing
and manipulation can be made simple and
foolproof. Here's how: Two different situa-
tions will be covered. In the first situation,
the requirement is straightforward: All
bytes of the run-time parameter must be
present for every invocation of the program.

Note that in Figure 3 if program main-
tenance ever changes the length of the
PICTURE clause of PARM-VALUE, the
IF statement remains valid. 

In the second situation, any parameter
length between one and 12 is considered
valid. However, the following three additional
requirements are imposed:

● a missing parameter must have a unique
error message

● a parameter value cannot exceed the
maximum allotted length

● no unprintable "core garbage" can be
moved into Working-Storage

TIP #2 SUMMARY

EDIT-FIELD1 and EDIT-FIELD2, as
shown in Figure 4, would both have a PIC-
TURE clause of Z,ZZ9. Note that the
LENGTH OF special register is valid as the

sending field in a MOVE as shown in the
move to EDIT-FIELD2.

The reference modification move that is
executed when the parameter length is
acceptable ensures that no unprintable data
gets moved into Working-Storage.

When a reference modification is named
in the sending field and the receiving field
does not contain a reference modification,
normal COBOL character move rules
apply; that is, left-justification with space
padding on the right in the receiving field.

TIP #3: WHEN-COMPILED

The WHEN-COMPILED special register
first appeared in COBOL II. It has been
upgraded in COBOL for MVS and can now
be used in a FUNCTION move. When used
as a FUNCTION, the WHEN-COMPILED
special register returns a four-position year
within the returned text. 

The WHEN-COMPILED special register
is, as its name implies, the date and time that
the program was compiled. For many years
now, I have been displaying an edited version

Working-Storage Section.
01  WS-Value            Pic  X(12) Value Spaces.

Linkage Section.
01  Passed-Parm-From-JCL.

05  Parm-Length     Pic S9(04) Binary.
05  Parm-Value      Pic  X(12).

FIGURE 2: COBOL LINKAGE AND WORKING-STORAGE
FOR A RUN-TIME "PARM=" DEFINITION

Working-Storage Section.
01  Date-Compiled-1.                                             

05  DC-Year Pic X(04).                          
05  DC-Month Pic X(02).                          
05  DC-Day Pic X(02).                          
05  DC-Hour Pic X(02).                          
05  DC-Minute Pic X(02).                          
05  DC-Second Pic X(02).                          

01  Display-Date-Compiled.                                       
05  DC-Month Pic X(02).                          
05 Pic X(01) Value '/'.                
05  DC-Day Pic X(02).                          
05 Pic X(01) Value '/'.                
05  DC-Year Pic X(04).                          
05 Pic X(03) Value ' @ '.              
05  DC-Hour Pic X(02).                          
05 Pic X(01) Value ':'.                
05  DC-Minute Pic X(02).                          
05 Pic X(01) Value ':'.                
05  DC-Second Pic X(02).  

Procedure Division.
Move Function When-Compiled to Date-Compiled-1.              
Move CORR Date-Compiled-1   to Display-Date-Compiled.        
Display "Program COBOLMVS compiled on "                      

Display-Date-Compiled " Beginning Execution ".  

FIGURE 5: ACQUIRING, EDITING AND DISPLAYING THE WHEN-COMPILED LITERAL

This edit ensures that a) the parameter is not missing, b) it's length does not exceed the allotted size and c) no unprint-
able, trailing characters get moved into Working-Storage.

If Parm-Length = Zero
Display Missing Parm Error and exit

Else
If Parm-Length > Length Of Parm-Value

Move Parm-Length         To Edit-Field1
Move Length Of Parm-Value To Edit-Field2
Display 'PARM= value contains too many characters' 
Display 'Length is ' Edit-Field1 ' Max Length is ' Edit-Field2
exit

Else
Move Parm-Value(1:PARM-LENGTH)To WS-Value.

FIGURE 4: AN EDIT TO ENSURE PROPER LENGTH FOR A PARAMETER

If Parm-Length NOT = Length Of Parm-Value
Display Error and exit

Else
Move Parm-Value to WS-Value.

FIGURE 3: AN EDIT TO ENSURE THAT ALL BYTES
OF THE RUN-TIME PARAMETER ARE PRESENT

of the WHEN-COMPILED literal as soon as
a COBOL program starts executing. This is
useful in debugging situations where it is
possible to confuse multiple versions of the
same program in different libraries.

Figure 5 shows a code sample of the
FUNCTION move along with a quick and
easy way to get the returned text into an
edited display format.

CONCLUSION

The overriding fact that I keep foremost in
my mind when hunting down ways to make
my programming easier and more accurate
is this: The feature is in the language for a
reason. Now go find that reason and use it to
maximum advantage!  

